最大値の最小値 with *Mathematica*

ー東大入試を *1行で* 解くー

2018 年12月 2日　Cabri研究会　生越　茂樹

「最大値の最小値」を求めるときのポイント.（「最小値の最大値」も同様.）
以下，　$y = f_a(x)$ について. 少なくとも3つの方法がある. (個人的にはC→B→Aの順に発見した.)
プログラムはA<B<Cの順に長い. 速度はC>B>A,C>A>B,A>B>C,A>C>B などまちまち.

方法A. **「 $f_a(x)$ の最大値の最小値≧k⇔∀a{∃x $f_a(x)$≧k} 」**
「 $f_a(x)$ の最小値の最大値≦k⇔∀a{∃x $f_a(x)$≦k} 」
或いは否定を取って
「 $f_a(x)$ の最大値の最小値≦k⇔∃a{∀x $f_a(x)$≦k} 」
「 $f_a(x)$ の最小値の最大値≧k⇔∃a{∀x $f_a(x)$≧k} 」

さらに
「 $f_a(x)$ の最大値の最大値≦k⇔∀a{∀x $f_a(x)$≦k} 」
「 $f_a(x)$ の最小値の最小値≧k⇔∀a{∀x $f_a(x)$≧k} 」
或いは否定を取って
「 $f_a(x)$ の最大値の最大値≧k⇔∃a{∃x $f_a(x)$≧k} 」
「 $f_a(x)$ の最小値の最小値≦k⇔∃a{∃x $f_a(x)$≦k} 」

∃,∀と不等号の並べ方が 2^3=8通りあり，その全てが最大，最小値と対応している.

方法B.「$f_a(x)$ の最大値の最小値が≧k⇔∃{x,a} ($f_a(x)$==k&&∀x　$f_a(x)$≦k) 」を使う.

【注】後半の「∀x　$f_a(x)$≦k」はaを固定して考えていて kがあるaについて最大値である事を表す.
∃{x,a} ($f_a(x)$==k)&&(∀x　$f_a(x)$≦k) とすると，∃{x,a} ($f_a(x)$==k)　を考える時のa の値と,
(∀x　$f_a(x)$≦k) のa の値が異なるので駄目.

方法C.「最大値M_a の計算」と「M_a の最小値の計算」の2段階に分ける.
最大値M_a　はきちんと2式(∀x $f_a(x)$≦M と ∃x $f_a(x)$≦M)で定義しその間を&&で結ぶ.
(この場合は 2式に分けても，1式にまとめても同じ)
M_a の最小値を求めるときは存在記号を使った方が解り易い.
即ち ∃{*a*, *M*} *M* == *k* && (上のステップで求めたMの条件式). ←∃{a} としないこと.

【注】もし全称記号を用いるなら，∀a ∃M M≧k&&(Mの条件式) とする.
「　Mの最小値が1以上である事を示せ」のような時は ∀a (Mの条件式)⇒ M≧1 とできる.
(∀a (Mの条件式)&&M≧1　は駄目.通常はFalse になる)
これを　∀a (Mの条件式)⇒ M≧k　とすると,「Mの条件式が偽」または「M≧k」となりうまく行かない.

In[21]:=

「最大値の最小値」（3つの中の最大値）

図の三角形ABCを底面とし，平面ABC上の点Hから
長さ4のポールHDを垂直に立てて家を建てる.
木の値段は 長くなると指数関数的に上がるので
3本の梁 AD,BD,CD の長さの最大値を最小にしたい.
その梁の長さを求めよ．(自作)





f_a (x) の最大値の最小値 ≧ k ⇔ ∀ a {∃ x f_a (x) ≧ k}
f_a (x) の最大値の最小値 ≦ k ⇔ ∃ a {∀ x f_a (x) ≦ k}

In[22]:=
```
a = {5, 0}; b = {0, 12}; c = {-9, 0}; H = {x, y};
dis[x_] := Total[x * x]; dis[x_, y_] := dis[x - y]; (*dis=distance^2*)
```

$\mathrm{dis}[a,\ H] \equiv (x-5)^2 + y^2,\ \mathrm{dis}[b,\ H] \equiv x^2 + (y-12)^2,\ \mathrm{dis}[c,\ H] \equiv (x+9)^2 + y^2$
の最大値M(x,y) の最小値を求めればよい.

In[24]:=
```
ForAll[{x, y}, (dis[a, H] ≥ m) ∨ (dis[b, H] ≥ m) ∨ (dis[c, H] ≥ m)];
Resolve[%, Reals] // Timing
```

Out[25]= $\left\{0.15625,\ m \le \frac{4225}{64}\right\}$

In[26]:=
```
Exists[{x, y}, (dis[a, H] <= m) ∧ (dis[b, H] <= m) ∧ (dis[c, H] <= m)];
Resolve[%, Reals] // Timing
```

Out[27]= $\left\{0.125,\ m \ge \frac{4225}{64}\right\}$

「最大値の最小値」（無数のxに対する最大値）

k を実数の定数とするとき, x の関数

$$f_k(x) = |x^3 - 3kx| \quad (-1 \leq x \leq 1)$$

の最大値を $M(k)$ で表す.
k が実数全体を動くとき $M(k)$ が最小となる k の値
および $M(k)$ の最小値を求めよ.　(1977年文理)

f_a (x) の最大値の最小値 ≧ k ⇔ ∀ a {∃ x f_a (x) ≧ k}
f_a (x) の最大値の最小値 ≦ k ⇔ ∃ a {∀ x f_a (x) ≦ k}

In[75]:=
```
ClearAll["Global`*"];
```

【方法A】

In[91]:=
```
ForAll[k, Exists[x, -1 ≤ x ≤ 1 && Abs[x^3 - 3 k x] ≥ m]];
Resolve[%, Reals] // Timing
```

Out[92]= $\left\{0.09375,\ m \leq \frac{1}{4}\right\}$

In[93]:=
```
ClearAll["Global`*"];
Exists[k, ForAll[x, -1 ≤ x ≤ 1 ⇒ Abs[x^3 - 3 k x] <= m]];
Resolve[%, Reals] // Timing
```

Out[95]= $\left\{0.09375,\ m \geq \frac{1}{4}\right\}$

「最大値の最小値」

$f(x)=\left|x^2+ax+b\right|$ $(-1\leq x\leq 1)$ の最大値を $M(a,b)$ とする.
a,b が任意の実数値をとる時, $M(a,b)$ の最小値を 求めよ.
(有名問題)

この問題は方法によって時間の違いが非常に大きく A(9秒) > B(12秒) ≫ C(200秒) の順に速い.

【方法A】

In[96]:=
```
Exists[{a, b}, ForAll[x, -1 ≤ x ≤ 1 ⇒ Abs[x^2 + a x + b] ≤ k]];
Resolve[%, Reals] // Timing
```

Out[97]= $\left\{8.65625,\ k\geq\frac{1}{2}\right\}$

「最小値の最大値」(図形問題)

xy平面上の点P(a,b) に対し，正方形 S(P) を連立方程式

$$|x-a| \leq \frac{1}{2},\ |y-b| \leq \frac{1}{2}$$

の表す領域として定め，原点と S(P) の点との最小値 を f(P) とする．点(2,1) を中心とする半径 1 の円周上を P が動くとき，f(P) の最大値を求めよ．(1996年文)



f_a (x) の最小値の最大値 ≤ k ⇔ ∀ a {∃ x f_a (x) ≤ k}
f_a (x) の最小値の最大値 ≥ k ⇔ ∃ a {∀ x f_a (x) ≥ k}

In[98]:=
```
ClearAll["Global`*"];
sqr := ImplicitRegion[Abs[x - a] ≤ 1 / 2 && Abs[y - b] ≤ 1 / 2, {x, y}];
```

【方法A】 A （7 秒）> C （20 秒）≫ B （75 秒） の順に速い.

In[110]:=
```
prop1 = ForAll[{a, b}, (a - 2)^2 + (b - 1)^2 == 1 ⇒ Exists[{x, y}, {x, y} ∈ sqr && x^2 + y^2 ≤ k]];
prop2 = Exists[{a, b}, (a - 2)^2 + (b - 1)^2 == 1 && ForAll[{x, y}, {x, y} ∈ sqr ⇒ x^2 + y^2 ≥ k]];
Resolve[prop1, Reals] // Timing
Resolve[prop2, Reals] // Timing
```

Out[112]= $\left\{7.64063,\ k \geq \frac{1}{2}\left(7+2\sqrt{10}\right)\right\}$

Out[113]= $\left\{6.20313,\ k \leq \frac{1}{2}\left(7+2\sqrt{10}\right)\right\}$

「最小値の最大値」(整数問題)

xy 平面上, x 座標, y 座標がともに整数であるような点(m,n) を格子点とよぶ．各格子点を中心として半径 r の円がえがかれており，傾き $\frac{2}{5}$ の任意の直線はこれらの円のどれかと共有点をもつという．このような性質を持つ実数 r の最小値を求めよ．(東大)





f_a (x) の最小値の最大値 ≤ k ⇔ ∀ a {∃ x f_a (x) ≤ k}
f_a (x) の最小値の最大値 ≥ k ⇔ ∃ a {∀ x f_a (x) ≥ k}

仮定より r の範囲は,　∀k ∃(m,n) $\frac{|2m-5n-k|}{\sqrt{29}} \leqq r$　となるから, QE では,

In[108]:=
```
ForAll[k, Exists[{m, n}, (m | n) ∈ Integers && Abs[2 m - 5 n - k] ≤ r]];
Resolve[%, Reals]
```

Out[109]= $\forall_{\{k\}} \exists_{\{m,n\}} ((m \mid n) \in \text{Integers} \&\& \text{Abs}[-k+2m-5n] \le r)$

しかし整数問題なので(上の出力の通り) QEはうまく動かない．そこで意味を考えてみると...
kを固定して，m, nが動くときの|2m-5n-k|の最小値を f (k) としたとき，「f (k) の最大値」を求めると良い．

(2m-5n)は任意の整数値を取れるので|2m-5n-k|の最小値 f(k)=min{x,1-x} (但し x はk の整数部分).
よって f(k)の最大値は1/2 . 故に 「$\frac{1}{2\sqrt{29}} \leqq r$」となる.

即ち，逆に論理式を「最小値の最大値」に直す事ができれば，そんなに難しくない．

結局「∀a∃x f(a,x)」型の問題は「最小値の最大値」などに翻訳するのが定石か？(私は知らなかったが...)
逆に「最小値の最大値」問題は「∀a∃x f(a,x)」型の問題に書き直すと少し難しく見える.

xy平面上の点P(a,b) に対し，正方形 S(P) を連立方程式

$$|x-a| \le \frac{1}{2},\quad |y-b| \le \frac{1}{2}$$

の表す領域として定め, 原点を中心とする半径 r の円とその内部をD とする．点(2,1) を中心とする半径 1 の円周上をP が動く時，任意のPに対して，D と S(P) が共有点を持つ r の最小値を求めよ. (1996年文 改)



● 実はこの問題で「最小値の最大値」が「∀a∃x」の形に書けることに気が付いた．私にとっては思い出深い問題.